
*
*           _________________________________________
*          |                                         |
*          |            BOTA              |\         |
*          |                          ____|+\        |
*          |____________|\  _  _     _|_  |  \_______|
*    VIN           |  - | \/ \/ \_   ///  |  /       |   ___
*      ___________/|\___| /\_/\_/ |_______|-/        |__|OUT|
*    _|__    |     |  + |/          INN2| |/ BOPA    |  |___|
*   /_   \   |     |                    |            |
*  // \   \  |     |                    |__||________|
*  \   \_//  |     |          RPZC      |  ||  CCOMP |
*   \____/   |     |                    |            |
*    _|_     |     |        _/\  /\  /\_|_/\  /\  /\_|
*    ///     |     |        |  \/  \/    RAV\/  \/
*            |     |        |_________________
*            |  ___|                          |
*            |  |                 CINT __||___|
*            |  |                     |  ||   |
*            |  |  |\ BCHOP   RINT    |       |OUT2
*            |  |__|+\                | |\    |
*            |     |  \____/\  /\  /\_|_|-\   |
*            |     |  /      \/  \/     |  \__|
*            |_____|-/             INN  |  /
*                  |/                 __|+/
*                                   _|_ |/
*                                   ///
*

.OPTIONS    GMIN = 1e-18

VIN         VIN   0    SIN( 0 50n 10k) AC 50n
BOTA        INN2  0    I  = 1*(V(VIN)-V(OUT))*.001
BOPA        OUT   0    V  = -V(INN2)*100000
CCOMP       INN2  OUT  20p
RAV         INN2  OUT  10000k
RPZC        INN2  OUT2 1k

BCHOP       VCHOP 0    V  = (V(VIN)-V(OUT))*100000
RINT        VCHOP INN  10000000000k
CINT        INN   OUT2 200u
BINT        OUT2  0    V  = -100000*V(INN) 

.control
set pensize = 2
ac          dec      1000       1m      100000k



let         vinput = vin - out
plot        db(mag(out/vinput))   ph(out) -ph(vinput)

alter       rint   resistance = 100000k
ac          dec      1000       1m       100000k
let         vinput = vin - out
plot        db(mag(out/vinput)) db(mag(out2/vinput))  ph(out) -ph(vinput)

alter       rint   resistance = 1000k
ac          dec      1000       1m       100000k
let         vinput = vin - out
plot        db(mag(out/vinput)) db(mag(out2/vinput))  ph(out) -ph(vinput)

alter       rint   resistance = 100k
ac          dec      1000       1m       100000k
let         vinput = vin - out
plot        db(mag(out/vinput)) db(mag(out2/vinput))  ph(out) -ph(vinput)

.endc

.end

=====================END_OF_SPICE============================

Start off with a gain limited OP Amp by itself.  



*           _________________________________________
*          |                                         |
*          |            BOTA              |\         |
*          |                          ____|+\        |
*          |____________|\  _  _     _|_  |  \_______|
*    VIN           |  - | \/ \/ \_   ///  |  /       |   ___
*      ___________/|\___| /\_/\_/ |_______|-/        |__|OUT|
*    _|__    |     |  + |/          INN2| |/ BOPA    |  |___|
*   /_   \   |     |                    |            |
*  // \   \  |     |                    |__||________|
*  \   \_//  |     |          RPZC      |  ||  CCOMP |
*   \____/   |     |                    |            |
*    _|_     |     |        _/\  /\  /\_|_/\  /\  /\_|
*    ///     |     |        |  \/  \/    RAV\/  \/
*            |     |        |_________________
*            |  ___|                          |
*            |  |                 CINT __||___|
*            |  |                     |  ||   |
*            |  |  |\ BCHOP   RINT    |       |OUT2
*            |  |__|+\                | |\    |
*            |     |  \____/\  /\  /\_|_|-\   |
*            |     |  /      \/  \/     |  \__|
*            |_____|-/             INN  |  /
*                  |/                 __|+/
*                                   _|_ |/
*                                   ///
*



Now change the integration such that the chopper
amp starts to contribute gain at frequency lower
than 1Hz.

Now change the frequency to 100Hz.

Now change the frequency to 1000Hz.



Now the two Op amps are acting like a single
one pole Op Amp. But really what is happening
is below 1kHz the chopper is dominating all
the gain while above 1kHz the higher frequency
Op Amp dominates the gain. 


